

Solving Difference equations with MAPLE

MAPLE can solve some difference equations using **rsolve** command. The command structure is **rsolve(eqns, fcns)** where

- The function **rsolve** attempts to solve the recurrence relation(s) specified in **eqns** for the functions in **fcns**, returning an expression for the general term of the function.
- The first argument should be a single recurrence relation or a set of recurrence relations and boundary conditions. Any expressions in **eqns** which are not equations will be understood to be equal to zero.
- The second argument **fcns** indicates the functions that **rsolve** should solve for. A function may be represented by a name or an unevaluated function call whose argument is a name. If the function is represented by a name, the index variable for the solution is deduced from the recurrence equations. If the function is represented by an unevaluated function call, the argument name is used as the solution index. If the solution index differs from the index name in a recurrence equation, then the equation must be independent of the solution index.
- If **fcns** is a set, the solution is returned as a set of equations. The left hand side of each equation will be the function name. Otherwise, an expression for the solution is returned.
- The procedure **rsolve** can currently solve linear recurrences with constant coefficients, systems of linear recurrences with constant coefficients, divide and conquer recurrences with constant coefficients, many first order linear recurrences, and some nonlinear first order recurrences.
- If insufficient boundary conditions are given, **rsolve** uses symbolic function names as default values. For linear recurrences with constant coefficients and systems of linear recurrences with constant coefficients, if no boundary conditions are given, symbolic function calls starting with index 0 are used.
- Boundary conditions for linear recurrences with constant coefficients and systems of linear recurrences with constant coefficients may be specified in one of three ways:

$$t(a)=c, \quad t(a..b)=c, \quad t(k=a..b)=f(k),$$

where **a** and **b** are integers, **c** is a constant with respect to the index variable for the sequence, **k** is a name, and **f(k)** is an expression of **k**. The boundary condition **t(a..b)=c** is equivalent to **t(a)=t(a+1)=...=t(b)=c** and the condition **t(k=a..b)=f(k)** is equivalent to **t(a)=f(a), t(a+1)=f(a+1), ..., t(b)=f(b)**,

First order linear difference equations

$$y_{n+1} + a y_n = b_n \quad \text{nonhomogeneous first order difference equation}$$

$$y_{n+1} + a y_n = 0 \quad \text{homogeneous first order difference equation}$$

General solution

$$y_n = y_{hom_n} + y_{p_n}$$

where

y_{hom_n} is the general solution of the homogeneous equation

y_{-p_n} is a particular solution of the nonhomogeneous equation

The case of homogeneous equation

$$y_{n+1} + a y_n = 0$$

If we want to find solution of the form $y_n = q^n$ then q must be a solution of the equation $q + a = 0$ which is called the characteristic equation. The general solution of the homogeneous equation is generated by the root of the characteristic equation $q = -a$.

General solution: $y_{hom_n} = c (-a)^n$

The constant c can be found from initial condition $y_0 = \alpha$

> **eq1:=y(n+1)+a*y(n)=0;**

$$eq1 := y(n+1) + a y(n) = 0$$

> **rsolve(eq1,y(n));**

$$y(0) (-a)^n$$

> **rsolve({eq1,y(0)=alpha},y(n));**

$$\alpha (-a)^n$$

The case of nonhomogeneous equation

General solution

$$y_n = y_{hom_n} + y_{-p_n}$$

$y_{hom_n} = y_0 (-a)^n$ is the general solution of the homogeneous equation

y_{-p_n} , the particular solution of the nonhomogeneous equation, generally, can be found using the variation of the constant method, we try to find a particular solution of the form

$y_{-p_n} = c_n (-a)^n$, we try to find the sequence (c_n) from the condition that (y_{-p_n}) to be a solution of the nonhomogeneous equation

So, $y_{-p_{n+1}} + a y_{-p_n} = b_n$

$$c_{n+1} (-a)^{(n+1)} - (-a) c_n (-a)^n = b_n$$

$$c_{n+1} - c_n = (-a)^{-(n+1)} b_n$$

$$n = 0 \quad c_1 - c_0 = (-a)^{(-1)} b_0$$

$$n = 1 \quad c_2 - c_1 = (-a)^{(-2)} b_1$$

.....

$$n = n - 2 \quad c_{n-1} - c_{n-2} = (-a)^{-(n-1)} b_{n-2}$$

$$n = n - 1 \quad c_n - c_{n-1} = (-a)^{(-n)} b_{n-1}$$

$$c_n - c_0 = \sum_{k=1}^{n-1} (-a)^{(-k)} b_{k-1}$$

choosing $c_0 = 0$, we can write

$$c_n = \sum_{k=1}^{n-1} (-a)^{(-k)} b_{k-1}$$

Thus, the particular solution has the following form

$$y_p = (-a)^n c_n = (-a)^n \left(\sum_{k=1}^{n-1} (-a)^{(-k)} b_{k-1} \right)$$

$$y_p = \sum_{k=1}^{n-1} (-a)^{(n-k)} b_{k-1}$$

> **eq1_nonhom := y(n+1) + a*y(n) = b(n);**

$$eq1_nonhom := y(n+1) + a y(n) = b(n)$$

> **rsolve(eq1_nonhom, y(n));**

$$y(0) (-a)^n + \left(\sum_{n0=1}^n (-a)^{(n-n0)} b(n0-1) \right)$$

Particular cases for (b_n)

1. $b_n = b$ for all n

> **eq1_nonhom := y(n+1) + a*y(n) = b;**

$$eq1_nonhom := y(n+1) + a y(n) = b$$

> **rsolve(eq1_nonhom, y(n));**

$$y(0) (-a)^n - \frac{b(-a)^n}{1+a} + \frac{b}{1+a}$$

2. $b_n = P_m(n)$ - polynomial of degree m with respect to n

> **eq1_nonhom := y(n+1) + a*y(n) = n^2 + n;**

$$eq1_nonhom := y(n+1) + a y(n) = n^2 + n$$

> **rsolve(eq1_nonhom, y(n));**

$$y(0) (-a)^n + \frac{2}{1 + 3a^2 + a^3 + 3a} + \frac{2(n+1) \left(\frac{n}{2} + 1 \right)}{1+a} + \frac{(-2a-4)(n+1)}{(1+a)^2} + \frac{2a(-a)^n}{1 + 3a^2 + a^3 + 3a}$$

3. $b_n = q^n P_m(n)$ - exponential multiplied with a polynomial of degree m with respect to n
if $q \neq -a$ we have

> **eq1_nonhom := y(n+1) + a*y(n) = q^n * n;**

$$eq1_nonhom := y(n+1) + a y(n) = q^n n$$

> **rsolve(eq1_nonhom, y(n));**

$$y(0) (-a)^n + \frac{q(-a)^n}{(q+a)^2} - \frac{(a+2q)q^n}{q^2+2aq+a^2} + \frac{(n+1)q^n}{q+a}$$

if $q = -a$ we have

> **eq1_nonhom := y(n+1) + a*y(n) = (-a)^n * n;**

$$eq1_nonhom := y(n+1) + a y(n) = (-a)^n n$$

> **rsolve(eq1_nonhom, y(n));**

$$y(0) (-a)^n - \frac{(-a)^n}{a} - \frac{(n+1) \left(\frac{n}{2} + 1 \right) (-a)^n}{a} + \frac{2(n+1)(-a)^n}{a}$$

Applications of the first order linear difference equation

1. Interest

A sum of money that is lent to a bank earns interest. Let us denote the amount of deposit after n time periods by S_n . The interest is denoted by I_n . The relation between S_n and I_n is given by

$$S_{n+1} = S_n + I_n$$

This formula states that the amount on deposit after n periods is equal to the sum of the amount on deposit after the previous period plus the interest earned. The question now is how the interest I_n is related to S_n .

Simple Interest. A simple way of offering interest is given by the option that the interest is proportional to the initial sum S_0 deposited, called the principal,

$$I_n = p S_0$$

Here, p is the constant interest rate (for example example, $p = 0.05$ means that there is an interest rate of 5%). Using this relation it results in the following first order difference equation for S_n ($n = 0, 1, \dots$),

$$S_{n+1} = S_n + p S_0$$

This equation represents a specific case of the first order difference equation where $a = -1$ and $b_n = p S_0$.

```
> int_eq:=S(n+1)=S(n)+p*S0;
                                int_eq := S(n + 1) = S(n) + p S0
> rsolve({int_eq,S(0)=S0},S(n));
                                S0 + p S0 (n + 1) - p S0
> simplify(%);
                                S0 + p S0 n
> factor(%);
                                S0 (1 + p n)
```

This formula is called the simple interest formula. According to this option, the amount on deposit after n periods is a linear function of the principal S_0 . It is assumed that n refers to the number of years.

Exercise: We are interested to know the number k of years needed to double any initial sum S_0 in the case of simple interest.

```
> sol:=rsolve({int_eq,S(0)=S0},S(n));
                                sol := S0 + p S0 (n + 1) - p S0
> s:=unapply(sol,n,S0);
                                s := (n, S0) -> S0 + p S0 (n + 1) - p S0
> eq:=2*S0=s(k,S0);
                                eq := 2 S0 = S0 + p S0 (k + 1) - p S0
> solve(eq,k);
                                1
                                p
```

Compound Interest. Another type of interest is given if interest is added to the principal at regular intervals, which are called conversion periods. For this case, the interest formula is given by

$$I_n = \frac{p S_n}{r}$$

Here, r refers to the number of conversion periods per year, this means $r = 12$ corresponds to a conversion period of one month, and $r = 4$ corresponds to a conversion period of one quarter. For this interest option, the difference equation for $S[n]$ is given by ($n = 0, 1, \dots$),

$$S_{n+1} = S_n + \frac{p S_n}{r} = \left(1 + \frac{p}{r}\right) S_n$$

This equation is an homogeneous first order linear difference equation, $-a = 1 + \frac{p}{r}$ and $b_n = 0$ for all n .

```
> int_eq_c:=S(n+1)=S(n)+p/r*S(n);
```

$$\text{int_eq_c} := S(n+1) = S(n) + \frac{p S(n)}{r}$$

> `sol:=rsolve({int_eq_c,S(0)=S0},S(n));`

$$\text{sol} := S0 \left(\frac{r+p}{r} \right)^n$$

Exercise: We are interested to know the number k of years needed to double any initial sum S_0 in the case of compound interest.

> `int_eq_c:=S(n+1)=S(n)+p/r*S(n);`

$$\text{int_eq_c} := S(n+1) = S(n) + \frac{p S(n)}{r}$$

> `sol:=rsolve({int_eq_c,S(0)=S0},S(n));`

$$\text{sol} := S0 \left(\frac{r+p}{r} \right)^n$$

> `s:=unapply(sol,n,S0);`

$$s := (n, S0) \rightarrow S0 \left(\frac{r+p}{r} \right)^n$$

> `eq:=2*S0=s(k,S0);`

$$\text{eq} := 2 S0 = S0 \left(\frac{r+p}{r} \right)^k$$

> `solve(eq,k);`

$$\frac{\ln(2)}{\ln\left(\frac{r+p}{r}\right)}$$

2. Loan Repayments

Loan repayments can be studied in a corresponding way. Let us consider the repayment of a housing loan that required any initial debt S_0 (the amount borrowed from a financial institution). The debt changes due to

- (i) repayments at regular intervals made to reduce the debt, and
- (ii) interest that has to be paid on the amount still owing.

By considering a compound interest, the debt S_{n+1} after $n+1$ payments is given by the following formula

$$S_{n+1} = \left(1 + \frac{p}{r} \right) S_n - R$$

Here, S_{n+1} and S_n are the debt after $n+1$ and n payments, respectively, and R is the constant

repayment. The term involving $\frac{p}{r}$ refers to the compound interest option.

Thus we get a nonhomogeneous first order difference equation with $-a = 1 + \frac{p}{r}$ and $b_n = -R$ for all n .

> **LR_eq:=S(n+1)=(1+p/r)*S(n)-R;**

$$LR_eq := S(n+1) = \left(1 + \frac{p}{r}\right) S(n) - R$$

> **sol:=rsolve({LR_eq,S(0)=S0},S(n));**

$$sol := S0 \left(\frac{r+p}{r}\right)^n - \frac{rR \left(\frac{r+p}{r}\right)^n}{p} + \frac{rR}{p}$$

The number k of conversion periods required to repay the debt can be calculated for this case on the basis of the solution formula.

> **s:=unapply(sol,n,S0,R);**

$$s := (n, S0, R) \rightarrow S0 \left(\frac{r+p}{r}\right)^n - \frac{rR \left(\frac{r+p}{r}\right)^n}{p} + \frac{rR}{p}$$

> **eq:=s(k,S0,R)=0;**

$$eq := S0 \left(\frac{r+p}{r}\right)^k - \frac{rR \left(\frac{r+p}{r}\right)^k}{p} + \frac{rR}{p} = 0$$

> **T:=solve(eq,k);**

$$T := \frac{\ln\left(\frac{rR}{-pS0 + rR}\right)}{\ln\left(\frac{r+p}{r}\right)}$$

3. The Pielou Logistic Equation (Riccati type)

$$x_{n+1} = \frac{\alpha x_n}{1 + \beta x_n}$$

> **P_eq:=x(n+1)=alpha*x(n)/(1+beta*x(n));**

$$P_eq := x(n+1) = \frac{\alpha x(n)}{1 + \beta x(n)}$$

> `sol:=rsolve({P_eq,x(0)=x0},x(n));`

$$sol := rsolve\left(\left\{x(n+1) = \frac{\alpha x(n)}{1 + \beta x(n)}, x(0) = x0\right\}, x(n)\right)$$

> `eq1:=subs(x(n)=1/y(n),x(n+1)=1/y(n+1),P_eq);`

$$eq1 := \frac{1}{y(n+1)} = \frac{\alpha}{y(n)\left(1 + \frac{\beta}{y(n)}\right)}$$

> `lin_eq:=y(n+1)=solve(eq1,y(n+1));`

$$lin_eq := y(n+1) = \frac{y(n) + \beta}{\alpha}$$


> `sol:=rsolve({lin_eq,y(0)=1/x0},y(n));`

$$sol := \frac{\left(\frac{1}{\alpha}\right)^n}{x0} + \frac{\beta}{-1 + \alpha} - \frac{\beta\left(\frac{1}{\alpha}\right)^n}{-1 + \alpha}$$

> `xx:=unapply(1/sol,n,alpha,beta,x0);`

$$xx := (n, \alpha, \beta, x0) \rightarrow \frac{1}{\frac{\left(\frac{1}{\alpha}\right)^n}{x0} + \frac{\beta}{-1 + \alpha} - \frac{\beta\left(\frac{1}{\alpha}\right)^n}{-1 + \alpha}}$$

> `plot([[j,xx(j,3,1,0.5)]$j=0..30],style=point,symbol=circle);`



> `alpha:=1;`

$$\alpha := 1$$

> `rsolve(lin_eq,y(n));`

$$y(0) + \beta (n + 1) - \beta$$


> `sol:=rsolve({lin_eq,y(0)=1/x0},y(n));`

$$sol := \frac{1}{x0} + \beta (n + 1) - \beta$$


```
> xx:=unapply(1/sol,n,beta,x0);
```

$$xx := (n, \beta, x0) \rightarrow \frac{1}{\frac{1}{x0} + \beta (n + 1) - \beta}$$

```
> plot([[j,xx(j,2,1)]$j=0..100],style=point,symbol=circle);
```



```
> alpha:='alpha';
```

$\alpha := \alpha$

Linear second order difference equation

$y_{n+2} + A y_{n+1} + B y_n = b_n$ nonhomogeneous second order difference equation

$y_{n+2} + A y_{n+1} + B y_n = 0$ homogeneous second order difference equation

General solution

$$y_n = c y_{hom_n} + y_{p_n}$$

where

y_{hom_n} is the general solution of the homogeneous equation

y_{p_n} is a particular solution of the nonhomogeneous equation

The case of homogeneous equation

$$y_{n+2} + A y_{n+1} + B y_n = 0$$

If we want to find solution of the form $y_n = q^n$ then q must be a solution of the equation

$$q^2 + A q + B = 0$$

which is called the characteristic equation. The general solution of the homogeneous equation is generated

by the roots of the characteristic equation.

I. If q_1, q_2 are real distinct roots then

$$y_{hom_n} = c_1 q_1^n + c_2 q_2^n$$

II. If $q_1 = q_2 = q$ double root then

$$y_{hom_n} = (c_1 + c_2 n) q^n$$

III. If $q_1 = r(\cos(\phi) + i \sin(\phi)), q_2 = r(\cos(\phi) - i \sin(\phi))$ are conjugated complex roots, written in the trigonometric form, then

$$y_{hom_n} = r^n (c_1 \cos(n\phi) + c_2 \sin(n\phi))$$

Coefficients c_1 and c_2 can be determined from initial conditions $y_0 = \alpha, y_1 = \beta$

The case of nonhomogeneous equation

General solution

$$y_n = y_{hom_n} + y_{p_n}$$

y_{hom_n} is the general solution of the homogeneous equation

y_{p_n} , the particular solution of the nonhomogeneous equation, generally, can be found only in some particular cases of b_n as in the case of the first order linear difference equation

Examples:

a) $y_{n+2} - 5y_{n+1} + 6y_n = 0$

b) $y_{n+2} - 4y_{n+1} + 4y_n = 0$

c) $y_{n+2} - 5y_{n+1} + 6y_n = 4n - 2$

d) $y_{n+2} - 5y_{n+1} + 6y_n = 4^n (an + b)$

> **eq_a:=y(n+2)-5*y(n+1)+6*y(n)=0;**

$$eq_a := y(n+2) - 5y(n+1) + 6y(n) = 0$$

> **ch_eq:=q^2-5*q+6=0;**

$$ch_eq := q^2 - 5q + 6 = 0$$

> **solve(ch_eq,q);**

3, 2

> **rsolve(eq_a,y(n));**

$$-(2y(0) - y(1))3^n - (-3y(0) + y(1))2^n$$

> **eq_b:=y(n+2)-4*y(n+1)+4*y(n)=0;**

$$eq_b := y(n+2) - 4y(n+1) + 4y(n) = 0$$

> **ch_eq:=q^2-4*q+4=0;**

$$ch_eq := q^2 - 4q + 4 = 0$$

> solve(ch_eq, q);

2, 2

> rsolve(eq_b, y(n));

$$-\left(\frac{1}{2}y(1) - 2y(0)\right)2^n + \left(-y(0) + \frac{1}{2}y(1)\right)(n+1)2^n$$

> eq_c := y(n+2) - 5*y(n+1) + 6*y(n) = 4*n - 2;

$$eq_c := y(n+2) - 5y(n+1) + 6y(n) = 4n - 2$$

> sol := rsolve(eq_c, y(n));

$$sol := -(2y(0) - y(1))3^n - (-3y(0) + y(1))2^n + 2n + 2 - 2 \cdot 2^n$$

> sol := rsolve({eq_c, y(0)=0, y(1)=1}, y(n));

$$sol := 3^n - 3 \cdot 2^n + 2n + 2$$

> eq_d := y(n+2) - 5*y(n+1) + 6*y(n) = 4^n * (a*n + b);

$$eq_d := y(n+2) - 5y(n+1) + 6y(n) = 4^n(2n + 4)$$

> sol := rsolve(eq_d, y(n));

$$sol := -(2y(0) - y(1))3^n - (-3y(0) + y(1))2^n + 4 \cdot 3^n - 5 \cdot 4^n + (n+1)4^n$$

> a:=2; b:=4;

$$a := 2$$

$$b := 4$$

> sol := rsolve({eq_c, y(0)=1, y(1)=0}, y(n));

$$sol := 2^n - 2 \cdot 3^n + 2n + 2$$

Applications of the linear second order difference equations

1. The Fibonacci Sequence (The Rabbit Problem)

This problem first appeared in 1202, in Liber abaci, a book about the abacus, written by the famous Italian mathematician Leonardo di Pisa, better known as Fibonacci. The problem may be stated as follows:

How many pairs of rabbits will there be after one year if starting with one pair of mature rabbits, if each pair of rabbits gives birth to a new pair each month starting when it reaches its maturity age of two months?

If $F(n)$ is the number of pairs of rabbits at the end of n months, then the recurrence relation that represents this model is given by the second-order linear difference equation

$$F(n+2) = F(n+1) + F(n), \quad F(0) = 1, \quad F(1) = 2$$

> Fib_eq := F(n+2) = F(n+1) + F(n);

$$Fib_eq := F(n+2) = F(n+1) + F(n)$$

> `sol_Fib:=rsolve({Fib_eq,F(0)=1,F(1)=2},F(n));`

$$sol_Fib := \left(\frac{3\sqrt{5}}{10} + \frac{1}{2}\right)\left(\frac{1}{2} + \frac{\sqrt{5}}{2}\right)^n + \left(\frac{1}{2} - \frac{3\sqrt{5}}{10}\right)\left(\frac{1}{2} - \frac{\sqrt{5}}{2}\right)^n$$

> `Fib_seq:=unapply(sol_Fib,n);`

$$Fib_seq := n \rightarrow \left(\frac{3\sqrt{5}}{10} + \frac{1}{2}\right)\left(\frac{1}{2} + \frac{\sqrt{5}}{2}\right)^n + \left(\frac{1}{2} - \frac{3\sqrt{5}}{10}\right)\left(\frac{1}{2} - \frac{\sqrt{5}}{2}\right)^n$$

> `Fib_seq(6);`

$$\left(\frac{3\sqrt{5}}{10} + \frac{1}{2}\right)\left(\frac{1}{2} + \frac{\sqrt{5}}{2}\right)^6 + \left(\frac{1}{2} - \frac{3\sqrt{5}}{10}\right)\left(\frac{1}{2} - \frac{\sqrt{5}}{2}\right)^6$$

> `expand(%);`

21

> `Fib_seq(12);`

$$\left(\frac{3\sqrt{5}}{10} + \frac{1}{2}\right)\left(\frac{1}{2} + \frac{\sqrt{5}}{2}\right)^{12} + \left(\frac{1}{2} - \frac{3\sqrt{5}}{10}\right)\left(\frac{1}{2} - \frac{\sqrt{5}}{2}\right)^{12}$$

> `expand(%);`

377

The Fibonacci sequence has many properties. One is $\lim_{n \rightarrow \infty} \frac{F(n+1)}{F(n)} = \frac{1+\sqrt{5}}{2}$, this number is called the *golden mean*, which supposedly represents the ratio of the sides of a rectangle that is most pleasing to the eye.

> `gm:=limit(Fib_seq(n+1)/Fib_seq(n),n=infinity);`

$$gm := \frac{1}{2} + \frac{\sqrt{5}}{2}$$

2. National Income Model

The national income represents a measure for the total level of economic activity within a country during a certain period of time. The economic activity of a country determines the living standard of people: a recession decreases and a boom increases the living standard. Thus, there is a great interest in the variations of the national income to understand, for example, the effect of factors that influence the national income. There is a variety of possibilities to define the national income and its influence factors. Here, we will focus on a relatively simple model pioneered by the American economist P. A. Samuelson (1939). Samuelson's model describes the national income y_n of a country during an accounting period $n = 0, 1, 2, \dots$, (n may refer to one year) as a sum of three contributions,

$$y_n = C_n + Inv_n + G_n$$

C_n is the amount spent on consumer goods (food, housing, clothing,...) during the accounting period.

Inv_n is the induced private investment (the amount invested in new machinery, equipment,...). G_n is the amount spent by the government (social security, health, education, infrastructure investments, research spending,...). To calculate y_n it is needed to relate C_n , Inv_n , and G_n to y_n . The consumption

expenditure C_n may be expected to be proportional to the national income of the preceding period (an increase in the living standard leads to an increase in consumption),

$$C_n = \alpha y_{n-1}$$

Here, the constant of proportionality $0 \leq \alpha$ and $\alpha \leq 1$ is called the marginal propensity to consume. The induced private investment Inv_n can be modeled in several ways. One possible assumption (which is called the acceleration principle) is that Inv_n is proportional to the increase in consumption of the period considered over the preceding period (an increasing living standard over the previous period leads to increased investments to offer additional goods),

$$Inv_n = \beta (C_n - C_{n-1}) = \alpha \beta (y_{n-1} - y_{n-2})$$

where the model parameter $0 \leq \beta$. Samuelson considered variations $0 \leq \beta$ and $\beta \leq 4$.

Finally, a simple model for the government expenditure G_n is that G_n is constant

$$G_n = \gamma$$

According to the three assumptions, Samuelson's national income model reads

$$y_n = \alpha y_{n-1} + \alpha \beta (y_{n-1} - y_{n-2}) + \gamma$$

thus

$$y_n = \alpha (1 + \beta) y_{n-1} - \alpha \beta y_{n-2} + \gamma$$

or

$$y_{n+2} = \alpha (1 + \beta) y_{n+1} - \alpha \beta y_n + \gamma$$

```
> Nat_Inc_eq:=y(n+2)=alpha*(1+beta)*y(n+1)-alpha*beta*y(n)+gamma;
Nat_Inc_eq := y(n + 2) = \alpha (1 + \beta) y(n + 1) - \alpha \beta y(n) + \gamma
```

```
> Nat_Inc_seq:=rsolve(Nat_Inc_eq,y(n)):
```

```
> char_eq:=q^2-alpha*(1+beta)*q+alpha*beta=0;
char_eq := q^2 - \alpha (1 + \beta) q + \alpha \beta = 0
```

```
> solve(char_eq,q);
```

$$\frac{\alpha}{2} + \frac{\alpha \beta}{2} + \frac{\sqrt{\alpha^2 + 2 \alpha^2 \beta + \alpha^2 \beta^2 - 4 \alpha \beta}}{2}, \frac{\alpha}{2} + \frac{\alpha \beta}{2} - \frac{\sqrt{\alpha^2 + 2 \alpha^2 \beta + \alpha^2 \beta^2 - 4 \alpha \beta}}{2}$$

The particular solution is a constant sequence (since $b_n = \gamma$ for all n), so $y_p = C$, the value of C is the solution of the equation

$$C = \alpha (1 + \beta) C - \alpha \beta C + \gamma$$

```
> eq:=C=alpha*(1+beta)*C-alpha*beta*C+gamma;
eq := C = \alpha (1 + \beta) C - \alpha \beta C + \gamma
```

```
> y_p:=solve(eq,C);
```

$$y_p := -\frac{\gamma}{\alpha - 1}$$

3. Propagation of Annual Plants

The material of this section comes from Edelstein–Keshet of plant propagation. Our objective here is to develop a mathematical model that describes the number of plants in any desired generation. It is known that plants produce seeds at the end of their growth season (say August), after which they die. Furthermore, only a fraction of these seeds survive the winter, and those that survive germinate at the beginning of the season (say May), giving rise to a new generation of plants. Let

γ = number of seeds produced per plant in August,
 α = fraction of one-year-old seeds that germinate in May,
 β = fraction of two-year-old seeds that germinate in May,
 σ = fraction of seeds that survive a given winter.

If p_n denotes the number of plants in generation n , then

$$p_n = \alpha sI_n + \beta s2_n$$

where

sI_n = the number of one-year-old seeds in April

$s2_n$ = the number of two-year-old seeds in April

Observe that the number of seeds left after germination may be written as
seeds left = (fraction of not germinated) x (original number of seeds in April)

thus

$$ssI_n = (1 - \alpha) sI_n$$

$$ss2_n = (1 - \beta) s2_n$$

where

ssI_n = the number of one-year-old not germinated seeds left in May

$ss2_n$ = the number of two-year-old not germinated seeds left in May

New seeds $s0_n$ (0-year-old) are produced in August at the rate of γ per plant,

$$s0_n = \gamma p_n$$

After winter, seeds $s0_n$ that were new in generation n will be one year old in the next generation $n+1$, and a fraction σ $s0_n$ of them will survive. Hence

$$sI_{n+1} = \sigma s0_n$$

or

$$sI_{n+1} = \sigma \gamma p_n$$

Similarly,

$$s2_n = \sigma ssI_n$$

which yields

$$s2_{n+1} = \sigma (1 - \alpha) sI_n$$

$$s2_{n+1} = \sigma^2 (1 - \alpha) \gamma p_{n-1}$$

Substituting for sI_n , $s2_n$ in expressions of p_n gives

$$p_n = \alpha sI_n + \beta s2_n$$

$$p_n = \alpha \sigma \gamma p_{n-1} + \beta \sigma^2 (1 - \alpha) \gamma p_{n-2}$$

or

$$p_{n+2} = \alpha \sigma \gamma p_{n+1} + \beta \sigma^2 (1 - \alpha) \gamma p_n$$

> **char_eq:=q^2-alpha*sigma*gamma*q-beta*sigma^2*(1-alpha)*gamma=0;**
char_eq := q² - α σ γ q - β σ² (-α + 1) γ = 0

> **solve(char_eq,q);**
 $\left(\frac{\alpha \gamma}{2} + \frac{\sqrt{\alpha^2 \gamma^2 - 4 \gamma \alpha \beta + 4 \beta \gamma}}{2} \right) \sigma, \left(\frac{\alpha \gamma}{2} - \frac{\sqrt{\alpha^2 \gamma^2 - 4 \gamma \alpha \beta + 4 \beta \gamma}}{2} \right) \sigma$

>